Crandall's Adjustment for a Closed Traverse

Programmer: Dr. Bill Hazelton

Date:	March, 2008.	Version: 1,0	Mnemonic: K for 'Krandall' (sorry)	
Line	Instruction	Display	User Instructions	
K001	LBL K		► LBL K	
K002	CLSTK		CLEAR 5	
K003	FS? 10		← FLAGS 3 .0	
K004	GTO K008			
K005	SF 1		← FLAGS 1 1	
K006	SF 10		← FLAGS 1 .0	
K007	GTO K009			
K008	CF 1		← FLAGS 2 1	
K009	CLΣ		► CLEAR 4	
K010	CRANDALL ADJ		(Key in using EQN RCL C, RCL R, etc.)	
K011	PSE		▶ PSE	
K012	100			
K013	STO I		► STO I	
K014	STO (I)		$rac{>}{>}$ STO (I) [(I) is on the 0 key]	
K015	ENTER FIX SIDES		(Key in using EQN RCL E, RCL N, etc.)	
K016	PSE		▶ PSE	
K017	ENTER FIX AZ		(Key in using EQN RCL E, RCL N, etc.)	
K018	PSE		▶ PSE	
K019	INPUT Q	Q?	SINPUT Q	
K020	RCL Q			
K021	x < 0?		r≥ x?0 3	
K022	GTO K036			
K023	$HMS \rightarrow$		► HMS→	
K024	STO Q		► STO Q	
K025	ENTER FIX DIST		(Key in using EQN RCL E, RCL N, etc.)	
K026	PSE		r≥ PSE	
K027	INPUT D	D?	SINPUT D	
K028	RCL Q			
K029	SIN			
K030	RCL× D			
K031	RCL Q			
K032	COS			
K033	RCL× D			
K034	Σ+			
K035	GTO K017			
K036	-30			
K037	STO I		STO I	
K038	0			
K039	STO (I)		$rac{1}{2}$ STO (I) [(I) is on the 0 key]	

Line	Instruction	Display	User Instructions
K040	-31		
K041	STO I		► STO I
K042	0		
K043	STO (I)		► STO (I)
K044	-32		
K045	STO I		► STO I
K046	0		
K047	STO (I)		STO (I)
K048	STO K		STO K
K049	20		
K050	STO I		► STO I
K051	21		
K052	STO J		► STO J
K053	ENTER VAR SIDES		(Key in using EQN RCL E, RCL N, etc.)
K054	PSE		▶ PSE
K055	ENTER VAR AZ		(Key in using EQN RCL E, RCL N, etc.)
K056	PSE		▶ PSE
K057	INPUT Q	Q?	SINPUT Q
K058	RCL Q		
K059	x < 0?		r≥ x?0 3
K060	GTO K082		
K061	$HMS \rightarrow$		► HMS→
K062	STO Q		STO Q
K063	STO (I)		STO (I)
K064	ENTER VAR DIST		(Key in using EQN RCL E, RCL N, etc.)
K065	PSE		▶ PSE
K066	INPUT D	D?	SINPUT D
K067	RCL D		
K068	STO (J)		$rac{P}{P}$ STO (J) [(J) is on the . key]
K069	RCL Q		
K070	SIN		
K071	Х		
K072	RCL Q		
K073	COS		
K074	RCL× D		
K075	Σ+		
K076	1		
K077	STO+ K		STO + K
K078	2		
K079	STO+ I		STO + I
K080	STO+ J		STO + J
K081	GTO K055		
K082	Σx^2		SUMS Σx^2
K083	Σy^2		SUMS Σy^2
K084	X		l
K085	Σxy		$rac{1}{2}$ SUMS Σxy

Line	Instruction	Display	User Instructions
K086	\mathbf{x}^2		$rac{1}{2}$ x ²
K087	—		
K088	STO D		STO D
K089	Σy		SUMS Σy
K090	Σxy		SUMS Σxy
K091	×		
K092	Σx		$ ightarrow$ SUMS Σx
K093	Σy^2		$rac{1}{2}$ SUMS Σy^2
K094	×		
K095	—		
K096	RCL÷ D		
K097	STO A		r STO A
K098	Σx		$rac{}{\sim}$ SUMS Σx
K099	Σxy		SUMS Σxy
K100	×		
K101	Σy		SUMS Σy
K102	Σx^2		$rac{}{\sim}$ SUMS Σx^2)
K103	×		
K104	—		
K105	RCL÷ D		
K106	STO B		► STO B
K107	0		
K108	STO V		► STO V
K109	STO L		r≥ STO L
K110	MISCLOSURE		(Key in using EQN RCL M, RCL I, etc.)
K111	PSE		▶ PSE
K112	LAT-NORTH		(Key in using EQN RCL L, RCL A, etc.)
K113	PSE		▶ PSE
K114	Σχ		$rac{1}{2}$ SUMS Σx
K115	STO N		► STO N
K116	VIEW N		S VIEW N
K117	DEP-EAST		(Key in using EQN RCL D, RCL E, etc.)
K118	PSE		▶ PSE
K119	Σy		$rac{1}{2}$ SUMS Σy
K120	STO E		► STO E
K121	VIEW E		SIEW E
K122	Σχ		$rac{1}{2}$ SUMS Σx
K123	x ²		$rac{1}{2}$ x ²
K124	Σy		$rac{1}{2}$ SUMS Σy
K125	X ²		\mathbf{r} \mathbf{x}^2
K126	+		
K127	$\sqrt{\mathbf{x}}$		
K128	STO D		STO D
K129	MISC DIST		(Key in using EQN RCL M, RCL I, etc.)
K130	PSE		PSE PSE
K131	VIEW D		SI VIEW D

K132 Ly Image: SUMS Expected and the system of the s	Line	Instruction	Display	User Instructions
K133 Σ_X Image: SUMS Σ_X K134 \div Image: SUMS Σ_X K135ATANImage: ATANK136 $x \ge 0?$ Image: ATANK137GTO K140Image: ATANK138360Image: ATANK140 \rightarrow HMSImage: ATANK141STO QImage: ATANK142MISC AZ(Key in using EQN RCL M, RCL I, etc.)K143PSEImage: ATANK144VIEW QImage: ATANK145ADJUSTED TRAV(Key in using EQN RCL A, RCL D, etc.)K146PSEImage: ATANK14720Image: ATANK148STO IImage: ATANK14921Image: ATANK150STO JImage: ATANK151RCL (J)Image: ATANK152STO QImage: ATANK154STO DImage: ATANK155X <> yK160RCL \times DK161RCL \times DK162RCL \times DK164RCL \times DK165RCL \times BK166RCL \times DK167STO QK168STO $+$ VK169SIDE AZK170PSEK171RCL QK172 \rightarrow HMSK173STO QK174VIEW QK174VIEW QK174VIEW QK175DSTK174VIEW QK175DSTK175DSTK175DSTK175DST<	K132	Σy		SUMS Σy
K134 K135ATANK135ATANK136 $x \ge 0$?K137GTO K140K138360K139+K140→HMSK141STO QK142MISC AZK144VIEW QK145ADJUSTED TRAVK146PSEK14720K148STO 1K14921K14921K150STO 1K151RCL (I)K152STO QK153RCL (I)K154STO DK155X <> yK161RCL × DK161RCL × DK162RCL × DK163RCL × DK164RCL × DK164RCL × DK165STO + VK166+K167STO QK168STO + VK169SIDE AZK171RCL QK172→HMSK173STO QK164RCL × DK165STO + VK168STO + VK169SIDE AZK171RCL QK171RCL QK172→HMSK173STO QK174VIEW QK175STO QK174VIEW QK175DST CORRNK176PSEK176PSEK175PSEK175PSEK175PSEK175PSEK175PSEK175PSE <td>K133</td> <td>Σx</td> <td></td> <td>$rac{1}{r}$ SUMS Σx</td>	K133	Σx		$rac{1}{r}$ SUMS Σx
K135 K136 K137 K138ATANK136 K137 K138 $\times 20^{\circ}$ K137 K138GTO K140K138 K139 K140 \rightarrow HMSK140 K141 \rightarrow HMSK141 K142MISC AZK142 K144 VIEW Q \square STO QK143 K144PSEK144 K147 20 \square PSEK145 K147 20 \square PSEK146 K148 K150 K151 K152 K157 K155 K156 K158 K167 K158 K161 RCL Q \square STO JK155 K163 K164 K164 K161 K163 K164 K164 K167 K166 K167 K166 K167 K166 K167 K166 K167 K170 PSE \square STO C STO QK166 K167 K170 K170 K170 K170 \square STO C STO VIEW Q K171 RCL QK166 K167 K172 STO Q \square STO C STO + V K169 K171 RCL QK174 K172 STO Q \square STO Q STO QK174 K172 STO Q \square STO Q STO C STO + V K169 K173 STO QK174 K172 STO Q \square STO Q STO QK174 K174 VIEW Q \square STO Q STO QK175 K175 K175 K175 \square CQ STO QK174 K175 STO Q \square STO Q STO QK174 K174 VIEW Q \square STO Q STO QK175 K175 STO Q \square STO Q STO QK175 STO Q \square STO Q STO Q	K134	÷		
K136 $x \ge 0?$ Image: Constraint of the second	K135	ATAN		ATAN
K137GTO K140K138360K139+K140 \rightarrow HMSK141STO QK142MISC AZK143PSEK144VIEW QK145ADJUSTED TRAVK146PSEK14720K148STO IK14921K14921K14021K14120K142STO IK143STO IK144STO IK145STO JK151RCL (I)K152STO QK153RCL (I)K154STO DK155 $x <> y$ K156SINK157×K161RCL× DK163 $x <> y$ K164RCL× DK165STO CK166+K167STO CK168STO+ VK169SIDE AZK170PSEK171RCL QK172 \rightarrow HMSK173STO QK174VIEW QK175STO QK173STO QK174VIEW QK174VIEW QK174VIEW QK175PSEK174VIEW QK174VIEW QK175PSEK174VIEW QK174VIEW QK174VIEW QK175PSEK176PSEK176PSEK176PSEK176PSE<	K136	$x \ge 0$?		r≥ x?0 5
K138 360 K139+K140 \rightarrow HMSK141STO QK142MISC AZK143PSEK144YIEW QK145ADJUSTED TRAVK146PSEK14720K148STO IK14921K150STO JK151RCL (I)K152STO QK153RCL (I)K154STO DK155 $x < > y$ K156SINK157 $x < > y$ K168RCL x AK164RCL x AK165RCL x AK166 $x < > y$ K166 $x < > y$ K167STO CK168SIO $x < y$ K169SIDE AZ K170PSEK171RCL QK172 \rightarrow HMSK173STO QK173STO QK174VIEW QK175DIST CORRNK176PSEK175DIST CORRNK176PSEK174VIEW QK175DIST CORRNK176PSEK174VIEW QK174VIEW QK175DIST CORRNK176PSEK176PSEK176PSEK176PSEK176PSEK176PSEK176PSEK176PSE<	K137	GTO K140		
K130 K140→HMSK141STO QSTO QK142MISC AZ(Key in using EQN RCL M, RCL I, etc.)K143PSEPSEK144VIEW QSTO VIEW QK145ADUSTED TRAV(Key in using EQN RCL A, RCL D, etc.)K146PSEPSEK14720STO IK148STO IImage: STO IK14921Image: STO JK150STO JImage: STO JK151RCL (I)Image: STO QK153RCL (J)Image: STO QK154STO DImage: STO DK155 $x <> y$ Image: STO DK158RCL QImage: STO DK160RCL× DImage: STO CK162RCL× DK163 $x <> y$ K164RCL× DK165SITO + VK166+K167STO QK168STO + VK169SIDE AZK170PSEK171RCL QK173STO QK174VIEW QK173STO QK174VIEW QK173STO QK174VIEW QK175DIST CORRNK176PSEK176PSEK176PSEK176PSEK176PSEK176PSEK176PSEK176PSEK176PSEK176PSEK176PSEK177STO CRRNK178	K138	360		
K140 \rightarrow HMSImage: HMSK141STO QImage: STO QK142MISC AZ(Key in using EQN RCL M, RCL I, etc.)K143PSEImage: STO QK144VIEW QImage: STO QK145ADJUSTED TRAV(Key in using EQN RCL A, RCL D, etc.)K146PSEImage: STO IK14720Image: STO IK148STO IImage: STO IK14921Image: STO IK150STO JImage: STO IK151RCL (I)Image: STO QK152STO QImage: STO QK153RCL (J)Image: STO QK154STO DImage: STO QK155x <> yImage: STO DK166SINImage: STO CK161RCL × DImage: STO CK162RCL × AK163x <> yK164RCL × DK165RCL × BK166+K167STO CK168STO + VK169SIDE AZK170PSEK171RCL QK172-HMSK173STO QK174VIEW QK175DIST CORRNK176PSEK176PSEK176PSEK175DIST CORRNK176PSEK176PSEK176PSEK176PSEK176PSEK176PSEK176PSEK176PSEK176PSE <td>K139</td> <td>+</td> <td></td> <td></td>	K139	+		
K141STO QImage: STO QK142MISC AZ(Key in using EQN RCL M, RCL I, etc.)K143PSEImage: PSEK144VIEW QImage: PSEK145ADJUSTED TRAV(Key in using EQN RCL A, RCL D, etc.)K146PSEImage: PSEK14720Image: PSEK148STO IImage: PSEK14921Image: PSEK150STO JImage: PSEK151RCL (I)Image: PSEK152STO QImage: PSEK153RCL (J)Image: PSEK154STO DImage: PSEK155 $x <> y$ Image: PSEK156SINImage: PSEK157 x Image: PSEK168RCL x DImage: PSEK160RCL x DImage: PSEK161RCL x DImage: PSEK162RCL x DImage: PSEK163RCL x DImage: PSEK164RCL x DImage: PSEK165STO $+ V$ Image: PSEK166 $+$ Image: PSEK172 $-HMS$ Image: PSEK173STO QImage: PSEK174VIEW QImage: PSEK175DIST CORRNImage: PSE	K140	→HMS		→HMS
K142MISC AZ(Key in using EQN RCL M, RCL I, etc.)K143PSEPSEK144VIEW Q \checkmark VIEW QK145ADJUSTED TRAV(Key in using EQN RCL A, RCL D, etc.)K146PSE \checkmark PSEK14720 \checkmark PSEK148STO I \checkmark STO IK14921 \checkmark STO JK150STO J \checkmark STO JK151RCL (I) \checkmark STO QK153RCL (J) \checkmark STO QK154STO D \checkmark STO DK155x <> y \checkmark STO DK156SIN \checkmark STO DK157 \times STO CK160RCL× DK161RCL× DK163RCL× DK164RCL× DK165RCL× BK166+K167STO CK168STO+ VK169SIDE AZK170PSEK171RCL QK172HMSK173STO QK174VIEW QK175DIST CORRNK175DIST CORRNK176PSEK176PSEK174VIEW QK174VIEW QK174VIEW QK174VIEW QK175DIST CORRNK176PSE	K141	STO Q		STO Q
K143PSEPSEK144VIEW Q \checkmark VIEW QK145ADJUSTED TRAV(Key in using EQN RCL A, RCL D, etc.)K146PSEImage PSEK14720Image PSEK148STO IImage PSEK14921Image PSEK150STO JImage PSEK151RCL (I)Image PSEK152STO QImage PSEK153RCL (I)Image PSEK154STO DImage PSEK155x <> yImage PSEK156SINImage PSEK167XImage PSEK168RCL × DImage PSEK164RCL × DImage PSEK165RCL × BImage PSEK166HImage PSEK167STO CImage PSEK168STO + VK169SIDE AZImage PSEK171RCL QImage PSEK172-HIMSImage PSEK174VIEW QImage PSEK174VIEW QImage PSEK174VIEW QImage PSEK174VIEW QImage PSEK175DIST CORRNImage PSEK176PSEImage PSEK175DIST CORRNImage PSEK175PSEImage PSE	K142	MISC AZ		(Key in using EQN RCL M, RCL I, etc.)
K144VIEW QKIWK145ADJUSTED TRAV(Key in using EQN RCL A, RCL D, etc.)K146PSEPSEK14720PSEK148STO IPSEK14921PSEK150STO JPSEK151RCL (I)PSEK152STO QPSEK153RCL (J)PSEK154STO DPSEK155 $x <> y$ PSEK156SINPSEK157XPSEK168RCL × DPSEK164RCL × DPSEK165STO + VPSEK166+PSEK170PSEPSEK171RCL QPSEK174VIEW QPSEK174VIEW QPSEK174VIEW QPSEK175DIST CORRNPSEK176PSEPSE	K143	PSE		PSE PSE
K145ADJUSTED TRAV(Key in using EQN RCL A, RCL D, etc.)K146PSEPSEK14720PSEK148STO IPSEK14921PSEK150STO JPSEK151RCL (J)PSEK153RCL (J)PSEK154STO DPSEK155 $x <> y$ PSEK156SINPSEK167RCL X DPSEK161RCL X DPSEK164RCL X DPSEK165STO Y PSEK166 $+$ PSEK167STO C PSEK168STO Y VPSEK170PSEPSEK171RCL Q PSEK173STO Q PSEK173STO Q PSEK174VIEW Q PSEK175DIST CORRNPSEK175DIST CORRNPSEK175DIST CORRNPSEK175DIST CORRNPSEK175DIST CORRNPSEK175DIST CORRNPSEK175DIST CORRNPSEK176PSEPSEK175DIST CORRNPSEK176PSEPSEK175DIST CORRNPSEK175DIST CORRNPSEK175DIST CORRNPSEK175STO CORRNPSEK175STO CORRNPSEK175STO CORRNPSEK175STO CORRN <td>K144</td> <td>VIEW Q</td> <td></td> <td>STINEW Q</td>	K144	VIEW Q		STINEW Q
K146PSEPSEK14720PSEK148STO ISTO IK14921PSEK150STO JPSEK151RCL (I)RCL (I)K152STO QPSEK153RCL (J)PSEK154STO DPSEK155 $x <> y$ K156SINK157 \times K158RCL QK160RCL× DK161RCL× AK162RCL× AK164RCL× BK166+K167STO CK168STO+ VK169SIDE AZK171RCL QK172 \rightarrow HMSK173STO QK174VIEW QK175DIST CORRNK176PSEK175DIST CORRNK176PSEK175DIST CORRNK176PSEK175PSEK175PSEK175PSEK175PSEK176PSEK175PSEK176PSEK175PSEK175PSEK175PSEK175PSEK175PSEK175PSEK175PSEK175PSEK175PSEK175PSEK175PSEK175PSEK175PSEK175PSEK175PSEK175PSEK175 <td>K145</td> <td>ADJUSTED TRAV</td> <td></td> <td>(Key in using EQN RCL A, RCL D, etc.)</td>	K145	ADJUSTED TRAV		(Key in using EQN RCL A, RCL D, etc.)
K14720K148STO IK14921K150STO JK151RCL (I)K152STO QK153RCL (J)K154STO DK155 $x <> y$ K156SINK157×K158RCL QK160RCL× DK161RCL× AK162RCL× AK164RCL× DK165RCL× BK166+K167STO CK168STO+ VK169SIDE AZK171RCL QK172 \rightarrow HMSK173STO QK174VIEW QK175DIST CORRNK175DIST CORRNK176PSEK175PSEK175PSEK175PSEK175PSEK176PSEK177PSEK178PSEK174VIEW QK175PSEK175PSEK175PSEK176PSEK176PSEK177PSEK178PSEK174VIEW QK175PSEK175PSEK175PSEK175PSEK175PSEK175PSEK175PSEK175PSEK175PSEK175PSEK175PSEK175PSEK175PSEK175PSE <td>K146</td> <td>PSE</td> <td></td> <td>PSE</td>	K146	PSE		PSE
K148STO 1FSTO 1K14921FSTO 1K150STO JFSTO JK151RCL (J)FSTO QK153RCL (J)FSTO QK154STO DFSTO DK155 $x <> y$ FSTO DK156SINFSTO DK157 x FSTO DK158RCL QFSTO DK160RCL× DFSTO CK161RCL× DFSTO CK164RCL× DFSTO + VK165RCL× BFSTO + VK166+FSTO + VK166STO+ VFSTO CK170PSEFPSEK171RCL QFSTO QK172 \rightarrow HMSFSTO QK173STO QFSTO QK174VIEW QFSTO QK175DIST CORRNFPSEK175DIST CORRNFPSE	K147	20		
K14921K150STO JK151RCL (I)K152STO QK153RCL (J)K154STO DK155 $x <> y$ K156SINK157 \times K158RCL QK160RCL× DK161RCL× DK162RCL× AK164RCL× DK165STO CK166+K166+K166+K166+K167STO CK168STO+ VK169SIDE AZK170PSEK171RCL QK172 \rightarrow HMSK173STO QK174VIEW QK175DIST CORRNK176PSEK175PSEK175PSEK175PSEK175PSEK176PSEK177PSEK178PSEK179PSEK170PSEK171PSEK172 \rightarrow HMSK175PSEK175PSEK175PSEK175PSEK176PSEK175PSEK176PSEK177PSEK175PSEK176PSEK176PSEK176PSEK176PSEK176PSEK176PSEK176PSEK176PSEK176PSEK1	K148	STO I		STO I
K150STO JSTO JK151RCL (I) \blacksquare K152STO Q \blacksquare K153RCL (J) \blacksquare K154STO D \blacksquare K155 $x <> y$ \blacksquare K156SIN \blacksquare K157 \times \blacksquare K158RCL Q \blacksquare K160RCL× D \blacksquare K161RCL× D \blacksquare K162RCL× A \blacksquare K163 $x <> y$ \blacksquare K164RCL× D \blacksquare K165RCL× B \blacksquare K166 $+$ \blacksquare K167STO C \blacksquare K168STO+ V \blacksquare K170PSE \blacksquare K171RCL Q \blacksquare K172 \rightarrow HMS \blacksquare K173STO Q \blacksquare K174VIEW Q \blacksquare K175DIST CORRN(Key in using EQN RCL D, RCL I, etc.)K176PSE \blacksquare K175DIST CORRN(Key in using EQN RCL D, RCL I, etc.)	K149	21		
K151RCL (I)K152STO QK153RCL (J)K154STO DK155 $x <> y$ K156SINK157 \times K158RCL QK159COSK160RCL× DK161RCL× DK162RCL× AK163 $x <> y$ K164RCL× BK165RCL× BK166+K166+K167STO CK168STO+ VK169SIDE AZK170PSEK171RCL QK172 \rightarrow HMSK173STO QK174VIEW QK175DIST CORRNK175DIST CORRNK176PSEK175DIST CORRNK176PSEK175DIST CORRNK176PSEK175DIST CORRNK176PSEK175DIST CORRNK176PSEK176PSEK176PSEK176PSEK176PSEK177DIST CORRNK175DIST CORRNK176PSEK175DIST CORRNK176PSE	K150	STO J		STO J
k152STO Q \blacksquare K153RCL (J) \blacksquare K154STO D \blacksquare K155 $x <> y$ \blacksquare K156SIN \blacksquare K157 \times \blacksquare K158RCL Q \blacksquare K159COS \blacksquare K160RCL× D \blacksquare K161RCL× D \blacksquare K162RCL× A \blacksquare K163 $x <> y$ \blacksquare K164RCL× D \blacksquare K165RCL× B \blacksquare K166+ \blacksquare K167STO C \blacksquare K168STO+ V \blacksquare K169SIDE AZ(Key in using EQN RCL S, RCL I, etc.)K170PSE \blacksquare K171RCL Q \blacksquare K173STO Q \blacksquare K174VIEW Q \blacksquare K175DIST CORRN(Key in using EQN RCL D, RCL I, etc.)K175DIST CORRN \blacksquare K176PSE \blacksquare K176PSE \blacksquare K176PSE \blacksquare K176PSE \blacksquare K176PSE \blacksquare K177FOR \blacksquare K178 \blacksquare K174VIEW Q \blacksquare K175DIST CORRN(Key in using EQN RCL D, RCL I, etc.)K176PSE \blacksquare K176PSE	K151	RCL (I)		
K153RCL (J)K154STO DK155 $x <> y$ K156SINK157 x K158RCL QK159COSK160RCL× DK161RCL× DK162RCL× AK163 $x <> y$ K164RCL× BK165RCL× BK166+K166+K167STO CK168STO+ VK169SIDE AZK170PSEK171RCL QK172 \rightarrow HMSK173STO QK174VIEW QK175DIST CORRNK176PSEFPSEFPSE	K152	STO Q		STO Q
K154S10 DS10 DK155 $x <> y$ K156K156SINK157K158RCL QK159K159COSK160K160RCL× DK161K161RCL× DK163K162RCL× AK164K163 $x <> y$ K164K164RCL× BK166K165RCL× BK166K166+FK167STO CFK168STO+ VFK169SIDE AZ(Key in using EQN RCL S, RCL I, etc.)K170PSEFK171RCL QFK172 \rightarrow HMSFK173STO QFK174VIEW QFK175DIST CORRNFK176PSEFSTO PSEFSTO QFK175DIST CORRNFK176PSEK176PSEK176PSEK176PSEK176PSEK176PSEK176PSEK176PSEK176PSEK175DIST CORRNK176PSEK176PSEK176PSEK176PSEK176PSEK176PSEK176PSEK176PSEK176PSEK176PSEK176PSEK176PSEK176PSEK176PSE	K153	RCL (J)		
K155 $x <> y$ K156SINK157×K158RCL QK159COSK160RCL× DK161RCL× DK162RCL× AK163 $x <> y$ K164RCL× BK165RCL× BK166+K167STO CK168STO+ VK169SIDE AZK170PSEK171RCL QK172→HMSK173STO QK174VIEW QK175DIST CORRNK176PSEF25F25EF25F25EF25F25E	K154	STO D		
K156SINK157×K158RCL QK159COSK160RCL× DK161RCL× DK162RCL× AK163 $x <> y$ K164RCL× DK165RCL× BK166+K167STO CK168STO+ VK169SIDE AZK170PSEK171RCL QK172→HMSK173STO QK174VIEW QK175DIST CORRNK176PSEF2PSE	K155	X <> y		-
K157×K158RCL QK158RCL QK159COSK160RCL× DK161RCL× DK162RCL× AK163 $x <> y$ K164RCL× DK165RCL× BK166+K166+K167STO CK168STO+ VK169SIDE AZK170PSEK171RCL QK172 \rightarrow HMSK173STO QK174VIEW QK175DIST CORRNK176PSEPSE	K150	SIN		-
K138RCL QK159COSK160RCL× DK161RCL× DK162RCL× AK163 $x <> y$ K164RCL× BK165RCL× BK166+K167STO CK168STO+ VK169SIDE AZK170PSEK171RCL QK172 \rightarrow HMSK173STO QK174VIEW QK175DIST CORRNK176PSEPSEPSER175PSEPSER175PSEPSER176PSEPSER175DIST CORRNK176PSEPSEK176PSER176P	K157	X		
K139COSK160RCL× DK161RCL× DK162RCL× AK163 $x <> y$ K164RCL× BK165RCL× BK166+K167STO CK168STO+ VK169SIDE AZK170PSEK171RCL QK172 \rightarrow HMSK173STO QK174VIEW QK175DIST CORRNK176PSEPSEPSER174PSEPSER175DIST CORRNK176PSEPSER176PSEPSEPSER176PSEPSER176PSER176PSER176PSER176PSER176PSER176PSER176PSER176PSER176PSER176PSER176PSER176PSER176PSER176PSER176PSER176PSER177PSER176PSER177PSER176PSER177PSER176PSER177PSER177PSER177PSER177PSER177PSER177PSER177PSER177PSER177PSE <td>K158</td> <td>KCL Q</td> <td></td> <td></td>	K158	KCL Q		
K100KCL× DK161RCL× DK162RCL× AK163 $x <> y$ K164RCL× DK165RCL× BK166+K167STO CK168STO+ VK169SIDE AZK170PSEK171RCL QK172 \rightarrow HMSK173STO QK174VIEW QK175DIST CORRNK176PSEF25PSEF35PSEF35F350 QK174PSEF35F350 QK175DIST CORRNK176PSEF35F350F35F350F350F350F350F350 QK174VIEW QK175DIST CORRNK176PSEF350F350F350F350F350F350F350F350 QK176F350F350F350F350F350F475DIST CORRNK176F350F476F350F476F350F476F350F476F350F476F350F476F350F476F350F476F350F476F350F476F350F476F350F476F350F476F350F476F350F477F470F477F470F477F470 <td>K139</td> <td></td> <td></td> <td>-</td>	K139			-
K101KCL× DK162RCL× AK163 $x <> y$ K164RCL× DK165RCL× BK166+K167STO CK168STO+ VK169SIDE AZK170PSEK171RCL QK172 \rightarrow HMSK173STO QK174VIEW QK175DIST CORRNK176PSEF29EF29EF29EF29EF302F302F302F303F304F305	K100	RCLX D		-
K102KCLX AK163 $x <> y$ K164RCL× DK165RCL× BK166+K167STO CK168STO+ VK169SIDE AZK170PSEK171RCL QK172 \rightarrow HMSK173STO QK174VIEW QK175DIST CORRNK176PSEPSEPSERCL QSTO QK174VIEW QK175DIST CORRNK176PSEPSEPSESTO QK176PSEPSESTO PSESTO P	K101	RCL× D		-
K105 $X <> y$ K164RCL× DK165RCL× BK166+K167STO CK168STO+ VK169SIDE AZK170PSEK171RCL QK172 \rightarrow HMSK173STO QK174VIEW QK175DIST CORRNK176PSEPSE	K102 K162	KCLX A		-
K104RCL× DK165RCL× BK166+K167STO CK168STO+ VK169SIDE AZK170PSEK171RCL QK172 \rightarrow HMSK173STO QK174VIEW QK175DIST CORRNK176PSEF2PSE	K105	X < > y		-
K105RCLX BK106+K166+K167STO CK168STO+ VK169SIDE AZK170PSEK171RCL QK172 \rightarrow HMSK173STO QK174VIEW QK175DIST CORRNK176PSEF2PSE	K104	RCLX D PCLX P		-
K100 $+$ K167STO CK168STO+ VK169SIDE AZK170PSEK171RCL QK172 \rightarrow HMSK173STO QK174VIEW QK175DIST CORRNK176PSEF176PSE	K105			-
K107S10 CK108STO + VK169SIDE AZK170PSEK171RCL QK172 \rightarrow HMSK173STO QK174VIEW QK175DIST CORRNK176PSE	K167	+ STO C		
K106S10+ VK109SIDE AZK170PSEK171RCL QK172 \rightarrow HMSK173STO QK174VIEW QK175DIST CORRNK176PSE	K167	STO C		$rac{1}{2}$ STO \pm V
K100SIDE AL(Rey in using EQUARCE 5, RCE 1, etc.)K170PSE \overrightarrow{P} PSEK171RCL Q \overrightarrow{P} PSEK172 \rightarrow HMS \overrightarrow{P} \rightarrow HMSK173STO Q \overrightarrow{P} STO QK174VIEW Q \checkmark VIEW QK175DIST CORRN(Key in using EQN RCL D, RCL I, etc.)K176PSE \overrightarrow{P} PSE	K160	SIDE A7		(Key in using EON RCL S RCL L etc.)
K170ISEK171RCL QK172 \rightarrow HMSK173STO QK174VIEW QK175DIST CORRNK176PSEPSE	K170	PSF		$\rightarrow PSF$
K171KOD \swarrow K172 \rightarrow HMSK173STO QK174VIEW QK175DIST CORRNK176PSEPSE	K171	RCLO		
K172 Mino K173 STO Q K174 VIEW Q K175 DIST CORRN K176 PSE	K172	\rightarrow HMS		
K174 VIEW Q K175 DIST CORRN K176 PSE	K173	STO O		STO 0
K175 DIST CORRN (Key in using EQN RCL D, RCL I, etc.) K176 PSE	K174	VIEW O		VIEW O
K176 PSE PSE	K175	DIST CORRN		(Key in using EON RCL D RCL Letc.)
	K176	PSE		$\rightarrow PSE$
K177 VIEW C	K177	VIEW C		S VIEW C

Line	Instruction	Display	User Instructions
K178	RCL C		
K179	STO+ D		STO + D
K180	CORRECTED DIST		(Key in using EQN RCL C, RCL O, etc.)
K181	PSE		▶ PSE
K182	VIEW D		└── VIEW D
K183	RCL Q		
K184	$HMS \rightarrow$		← HMS→
K185	STO Q		► STO Q
K186	SIN		
K187	RCL× C		
K188	RCL Q		
K189	COS		
K190	RCL× C		
K191	Σ+		
K192	1		
K193	STO+ L		► STO + L
K194	RCL K		
K195	RCL L		
K196	$x \ge y$?		≤ 1 x?y 5
K197	GTO K202		
K198	2		
K199	STO+ I		► STO + I
K200	STO+ J		► STO + J
K201	GTO K151		
K202	SUM OF CORRNS		(Key in using EQN RCL S, RCL U, etc.)
K203	PSE		PSE PSE
K204	VIEW V		SIEW V
K205	ADJ MISCLOSURE		(Key in using EQN RCL A, RCL D, etc.)
K206	PSE		PSE PSE
K207	LAT-NORTH		(Key in using EQN RCL L, RCL A, etc.)
K208	PSE		PSE PSE
K209	Σχ		$rac{1}{2}$ SUMS Σx
K210	STO N		STO N
K211	VIEW N		ST VIEW N
K212	DEP-EAST		(Key in using EQN RCL D, RCL E, etc.)
K213	PSE		PSE PSE
K214	Σy		SUMS Σy
K215	STO E		STO E
K216	VIEW E		ST VIEW E
K217	Σx		ΣX
K218	X ²		
K219	Σy		$rac{1}{2}$ SUMS Σy
K220	X ²		\mathbf{r} \mathbf{x}
K221	+		4
K222	$\sqrt{\mathbf{x}}$		l
K223	STO D		STO D

Line	Instruction	Display	User Instructions
K224	MISC DIST		(Key in using EQN RCL M, RCL I, etc.)
K225	PSE		▶ PSE
K226	VIEW D		✓ VIEW D
K227	Σy		SUMS Σy
K228	Σx		$rac{}{\sim}$ SUMS Σx
K229	÷		
K230	ATAN		r ATAN
K231	$x \ge 0$?		r≥ x?0 5
K232	GTO K235		
K233	360		
K234	+		
K235	→HMS		→HMS
K236	STO Q		► STO Q
K237	MISC AZ		(Key in using EQN RCL M, RCL I, etc.)
K238	PSE		r≥ PSE
K239	VIEW Q		ST VIEW Q
K240	100		
K241	STO I		r≥ STO I
K242	0		
K243	STO (I)		STO (I)
K244	20		
K245	RCL I		
K246	x < y?		≤ x?y 3
K247	GTO K251		
K248	1		
K249	STO- I		► STO — (I)
K250	GTO K242		
K251	PROGRAM END		(Key in using EQN RCL P, RCL R, etc.)
K252	PSE		▶ PSE
K253	FS? 1		← FLAGS 3 1
K254	CF 10		← FLAGS 2 .0
K255	RTN		ST RTN

Crandall's Adjustment for a Closed Traverse

Notes

- (1) The program is designed to run in RPN mode. Its performance in ALG mode is unknown and may produce erroneous results.
- (2) The program can work with traverses where all the sides are to be adjusted, where some of the sides are considered 'fixed' (not to be adjusted), and traverses between known points with co-ordinates. The last need the azimuth and distance between the fixed points derived from the co-ordinates, and treated as a fixed side.
- (3) The sequence of operations is to enter the fixed sides (those not to be adjusted), and then the variable sides (those to be adjusted). The calculator stores the variable sides and retrieves them from memory to compute the adjusted lengths, and also re-computes the traverse misclosure. The memory limits the traverse to no more than 390 variable sides. If you want to run Crandall's Adjustment on a

Crandall's Adjustment for a Closed Traverse

traverse of more than 390 variable sides: (a) use a different computational device; (b) be aware that the errors in such a traverse may be too difficult to predict and work with in a meaningful way; (c) use a full least squares adjustment, plus use cross-ties on the ground; and (d) put your analyst on danger money!

- (4) Provided that all the fixed sides are entered first, and then all the variable sides, it does not matter in which order the sides within each group are entered.
- (5) Azimuths are entered in HP notation, i.e., DDD.MMSS
- (6) The misclosure components of the adjusted traverse in X (or E or departure) and Y (or N or latitude) can be displayed by recalling Σy and Σx using the SUMS menu. (Note these are back-to-front.)
- (7) This program was inspired by the program developed by the one-and-only Neil H. Bradbury, LS, in January, 1977, for the HP-25 calculator.
- (8) The program assumes that the error in the length of each side is proportional to the length of the side, which is the situation with EDM. It will produce somewhat different results to a Crandall's adjustment based on the error increasing as the square root of the distance, which is the more traditional method based on error propagation using a tape.
- (9) To indicate the end of the groups of sides of the traverse to be entered at any point, enter a negative value for the azimuth, -1 being easiest.
- (10) At the end of the program, Flag 10 is reset to its value before the program started, and the 'indirect' memory is cleared and so de-allocated. This saves memory space and is simply good practice. If you quit before the program comes to the very end, you may leave quite a lot of memory used (from location 20 up to 100).
- (11) This program does not use vectors or complex numbers for the misclosure calculation, because the components are required for the adjustment calculations. Consequently, the components are calculated and the vector sum is accumulated through the statistical registers, together with the various data items for the adjustment calculation.
- (12) To avoid a problem with storing an azimuth (or distance) of zero, which may cause a memory error in the 'indirect' registers, an arbitrary value is stored in register 100. This should allow a 40-sided traverse. If you want to try a bigger traverse that may have an azimuth of zero in it, change the value of 100 at lines K012 and K240 to a value that is a little larger than twice the number of sides in the traverse plus 20. And make yourself some strong coffee! This will allocate enough 'indirect' memory at the start of the program, and clear and de-allocate it when the program ends. This particular issue came up in discussion with Mr. James Watson of RMIT, and brought to light the potential for a very subtle error in memory allocation, which this addition to the program should allow the user to avoid.

Theory

Crandall's Adjustment was developed by Charles Crandall, and published in 1914. It was the first serious rival to Bowditch's Method of Adjustment. The fundamental concepts were that angle measurement was far better than distance measurement, and that the small random errors in the angles of a traverse could be properly adjusted by simple distribution of the angular misclosure among the traverse's angles. This meant that all the remaining misclosure in the traverse, i.e., all

Crandall's Adjustment for a Closed Traverse

the random errors, could be accounted for by the distances alone, so these were all that was adjusted by Crandall's Adjustment.

Crandall used least squares adjustment by condition equations for the adjustment. By avoiding having to adjust the angles as part of the adjustment proper, the number of conditions was reduced to two, which greatly simplified the calculations.

Angle Adjustment

The angular misclosure of the traverse can be adjusted by any reasonable means. The misclosure can be distributed evenly among all the angles, or larger amounts may be apportioned to angles that can be assumed to be less reliable, perhaps because of short lines or poor lines of sight. However the angles are adjusted, they should be adjusted to bring the misclosure to zero, and then the azimuths of each line calculated.

It is a good idea to run the traverse through one of the closure programs to make sure that there are no gross or systematic errors in the traverse, prior to adjustment. This is because the adjustment process assumes that only random errors are present in the data. *If this assumption is not correct, the adjustment will produce erroneous results.*

Distance Adjustment

With the angular misclosure adjusted to zero, there are now only two conditions that must be met to have a consistent traverse. (The third condition, that the angular misclosure is zero, has been met by adjusting the angle ahead of this point.) Solution of a 2×2 matrix is straightforward, so a full least squares adjustment by condition equations can be performed. This adjustment changes only the lengths of the sides of the traverse. The angles are already adjusted and so held fixed.

If the length of each line in the traverse is M_i , while θ_i is its azimuth, then d_i will be the adjusted length of each line in the traverse, and v_i will be the correction to the measured length of each line to obtain the adjusted length:

$$d_i = M_i + v_i \tag{1}$$

The conditions to be met can be expressed as:

$d_1 \cos \theta_1 + d_2 \cos \theta_2 + d_3 \cos \theta_3 + \dots = 0$	[Sum of latitudes equals zero]
$d_1 \sin \theta_1 + d_2 \sin \theta_2 + d_3 \sin \theta_3 + \dots = 0$	[Sum of departures equals zero]

Substituting Equation [1] into the above equations gives:

$$(M_1 + v_1)\cos\theta_1 + (M_2 + v_2)\cos\theta_2 + (M_3 + v_3)\cos\theta_3 + \dots = 0$$

$$(M_1 + v_1)\sin\theta_1 + (M_2 + v_2)\sin\theta_2 + (M_3 + v_3)\sin\theta_3 + \dots = 0$$

Re-arranging and gathering terms produces:

$$v_1 \cos \theta_1 + v_2 \cos \theta_2 + v_3 \cos \theta_3 + \dots + q_1 = 0$$

$$v_1 \sin \theta_1 + v_2 \sin \theta_2 + v_3 \sin \theta_3 + \dots + q_2 = 0$$
 [2]

where $q_1 = M_1 \cos \theta_1 + M_2 \cos \theta_2 + M_3 \cos \theta_3 + \dots =$ misclosure in latitude $q_2 = M_1 \sin \theta_1 + M_2 \sin \theta_2 + M_3 \sin \theta_3 + \dots =$ misclosure in departure

HP-35s Calculator Program Crandall's Adjustment for a Closed Traverse

So q_1 and q_2 are obtained directly from the closure check of the traverse, done after the angles are adjusted and azimuths computed. Representing the departure of each line by D_i , and the latitude of each line by L_i , Equations [2] can be rewritten as:

Closure 6A

$$v_{1} \cos \frac{L_{1}}{d_{1}} + v_{2} \cos \frac{L_{2}}{d_{2}} + v_{3} \cos \frac{L_{3}}{d_{3}} + \dots + q_{1} = 0$$

$$v_{1} \sin \frac{D_{1}}{d_{1}} + v_{2} \sin \frac{D_{2}}{d_{2}} + v_{3} \sin \frac{D_{3}}{d_{3}} + \dots + q_{2} = 0$$
[3]

The theory of least squares now allows the development of what are termed the normal equations (named after the normal distribution, which describes the distribution characteristics of the random errors). The details can be found elsewhere, but basically they are a combination of all the measured data, the precision values of the measurements, and the models that describe how they all fit together, i.e., the model of the conditions to be met (e.g., Equations [2] and [3]).

There are two cases that Crandall considered. The first is the situation where the error in the length of the sides increases in proportion to the square root of the length of the line, i.e., \sqrt{d} , which is a good assumption for taped distances. The second case is where the error in distance increases in proportion to the length of the line, which is more the case with short distances measured with rods, but also the case with EDM. This program is based on the second case, as it is more probable that EDM traverses will be used with this program.

If the error in the lengths of the sides of the traverse increases in proportion to the length of each side, then the normal equations for the adjustment are as follows:

$$A\sum_{i} L_{i}^{2} + B\sum_{i} L_{i}D_{i} + q_{1} = 0$$

$$A\sum_{i} L_{i}D_{i} + B\sum_{i} D_{i}^{2} + q_{2} = 0$$
[4]

Solving for A and B, which are called the correlatives:

$$A = \frac{q_{2}\sum_{i}L_{i}D_{i} - q_{1}\sum_{i}D_{i}^{2}}{\sum_{i}D_{i}^{2}\sum_{i}L_{i}^{2} - \left(\sum_{i}L_{i}D_{i}\right)^{2}}$$

$$B = \frac{q_{1}\sum_{i}L_{i}D_{i} - q_{2}\sum_{i}L_{i}^{2}}{\sum_{i}D_{i}^{2}\sum_{i}L_{i}^{2} - \left(\sum_{i}L_{i}D_{i}\right)^{2}}$$
[5]

the corrections to the line lengths can be calculated using:

$$v_i = L_i d_i A + D_i d_i B$$
[6]

Crandall's Adjustment for a Closed Traverse

As an aside, the corrections should all sum to zero if the traverse has no fixed sides, but that can be affected by rounding during the calculations, especially if the sums of the various components are large, so the result may be not quite zero. If the traverse has fixed sides, the sum of the corrections will be significantly different from zero, depending upon the length of the fixed side.

Including Fixed Sides and Points

In the event that the traverse includes more than one 'fixed' point (usually the starting point in calculations), Crandall's Method can still be used. A 'fixed' point means a point that has known co-ordinates, usually from a more precise prior survey, and will not be subject to adjustment in this program. This situation arises when a traverse connects to more than one known point.

If there are two fixed points, and these form the end points of one line of the traverse, then the traverse calculations are done in the normal way, except that the L_i , D_i , and L_iD_i values for the fixed line are not calculated and not included in the summations to solve the correlatives. This is the case in this program.

If there are more than two fixed points, or the fixed points are not adjacent, the traverse must be split into sections between fixed points, and the sections adjusted independently. This is not the ideal solution, as a full least squares adjustment should be used in this situation.

If the traverse runs between two known points, then the line (vector) between the two known points (computed from the co-ordinates of the two points) can be considered another side in the traverse. Computing around the traverse, the misclosure will reflect how well the traverse agrees with this computed vector. To adjust the traverse, don't include the L_i , D_i , and L_iD_i values for this line in the summations, and all the corrections will occur in the measured traverse lines.

For this program, any fixed lines should be worked out ahead of time, including the line between known end-points of the traverse (if applicable). The first part of the program allows the user to enter these sides, and they do not add to the data that sets up the adjustment. The second part of the program allows the user to enter the sides that do add to the adjustment. The final part of the program produces the results.

Sample Computations and Running the Program

1. All sides to be adjusted

The traverse to be adjusted is as follows:

Azimuth	Distance	
0° 12'	156.41	
90° 00'	211.65	
165° 49'	173.82	
250° 55'	176.60	
308° 30'	112.26	

Crandall's Adjustment for a Closed Traverse

The initial misclosure is: DE = 0.036 DN = 0.033Misclosure Length = 0.049 Misclosure Azimuth = 47° 24' 15"

Running the Program

Press XEQ K and then ENTER to start the program.

The program briefly displays CRANDALL ADJ, then briefly displays ENTER FIX SIDE to indicate that the fixed sides should be entered first.

The program then prompts ENTER FIX AZ, to indicate that the azimuth of a fixed side should be entered in DDD.MMSS (HP) notation. The prompt is to enter Q?.

As there are no fixed sides in this case, key in 1 then press the +/- key, and press R/S.

The calculator shows ENTER VAR SIDE to indicate that the sides to be varied (adjusted) should be entered now.

The calculator then prompts ENTER VAR AZ to indicate that the azimuth of a variable sides should be entered in DDD.MMSS (HP) notation. The prompts to enter is Q?.

Key in the azimuth of the first side, 0.12, then press R/S.

The calculator prompts with ENTER VAR DIST, to indicate that the length of that variable side should be entered. The prompt is to enter D?.

Key in the distance 156.41, then press R/S.

The calculator prompts with ENTER VAR AZ, then the Q? prompt.

Key in the next azimuth, 90.0, and press R/S.

The calculator prompts with ENTER VAR DIST, then the D? prompt.

Key in the distance, 211.65, and press R/S.

The calculator prompts with ENTER VAR AZ, then the Q? prompt.

Key in the next azimuth, 165.49, and press R/S.

The calculator prompts with ENTER VAR DIST, then the D? prompt.

Key in the distance, 173.82, and press R/S.

The calculator prompts with ENTER VAR AZ, then the Q? prompt.

Key in the next azimuth, 250.55, and press R/S.

The calculator prompts with ENTER VAR DIST, then the D? prompt.

Key in the distance, 176.6, and press R/S.

The calculator prompts with ENTER VAR AZ, then the Q? prompt.

Key in the next azimuth, 308.3, and press R/S.

The calculator prompts with ENTER VAR DIST, then the D? prompt.

Key in the distance, 112.26, and press R/S.

The calculator prompts with ENTER VAR AZ, then the Q? prompt.

Crandall's Adjustment for a Closed Traverse

As this is the last variable side, key in a negative azimuth, -1, and press R/S.

[Note that the Q and D prompts contain the previous entries, albeit the Q value is in decimal degrees, which can help you keep track of where you are when entering sides.]

The calculator briefly displays MISCLOSURE, then briefly displays LAT-NORTH, then stops and displays:

N= 0.0330

which is the misclosure component in the latitude, or in the north-south direction. Press R/S to continue. The calculator briefly displays DEP–EAST, then stops and displays:

E= 0.0359

which is the misclosure in the departure or east-west direction. Press R/S to continue. The calculator briefly MISC DIST, then stops and displays:

D=

0.0488

which is the length of the misclosure vector. Press R/S to continue. The calculator briefly displays MISC AZ, then stops and displays:

Q=

47.2415

which is the azimuth of the misclosure vector, in HP notation (DDD.MMSS). Press R/S to continue.

The calculator briefly displays ADJUSTED TRAV and then proceeds around a loop displaying, in turn, the azimuth of the side, the correction to that side's distance, and the corrected distance for the side. The prompts before each value are SIDE AZ and Q= for the traverse side's azimuth (which is displayed in HP notation, DDD.MMSS), DIST CORRN and C= for the correction to the distance, and CORRECTED DIST and D= for the corrected side distance.

Pressing R/S after each value is displayed will cause the calculator to move on to the next value. The results are tabulated below. The sequence is moving along the rows from left-to-right, and then down. Note that the azimuth of the sides are unchanged: only the side lengths have been adjusted.

Side Azimuth	Distance Correction	Corrected Distance
0° 12' 00"	-0.0139	156.3961
90° 00' 00"	-0.0207	211.6293
165° 49' 00"	0.0132	173.8332
250° 55' 00"	0.0194	176.6194
308° 30' 00"	0.0001	112.2601

Crandall's Adjustment for a Closed Traverse

After the last corrected side distance has been displayed, the calculator then briefly displays SUM OF CORRNS, which is for the total of the corrections. The calculator stops and displays:

which is the sum of all the distance corrections. This is pretty close to zero. Press R/S to continue. The calculator briefly displays ADJ MISCLOSURE, then LAT–NORTH briefly, then stops and displays:

which indicates that the misclosure of the adjusted traverse in the north-south direction, or in the latitudes, is exceedingly small. Press R/S to continue, and the calculator briefly displays DEP-EAST, then stops and displays:

which is the misclosure adjusted traverse in the east-west or departure direction, and is also very, very small. Press R/S to continue and the calculator displays MISC DIST briefly, then stops and displays:

which is the length of the misclosure vector of the adjusted traverse which is also exceedingly tiny. Press R/S to continue, and the calculator briefly displays MISC AZ, then stops and displays:

which is the azimuth of the misclosure vector of the adjusted traverse, in HP notation (DDD.MMSS). Press R/S to continue.

The calculator then shows RUNNING for several seconds as it clears the indirect memory, and then briefly displays PROGRAM END, and terminates.

You can see that it is handy to have the traverse data nicely tabulated, to make getting it into and out of the calculator quick, simple and error free.

2. One fixed side, a traverse between known points

For this traverse, one side is considered fixed, because the traverse is between two fixed points. The azimuth and distance between the two points are deduced from co-ordinates (using a different program), and are 59° 04' 33" for 1995.78. The angular misclosure is adjusted out of the traverse angles, and the azimuths of the sides calculated, on the same azimuth datum as the fixed side, i.e., the same co-ordinate system.

The program is started and the fixed side entered: 59.0433 and 1995.78.

HP-35s Calculator Program Crandall's Adjustment for a Closed Traverse

As this is the only fixed side, the next azimuth is entered as -1, and the calculator moves to collect the data for the variable sides. These are as follows:

A:	zimuth			Distance	
3	349° 47'	19"		429.20	
1	.09° 22'	12"		476.00	
2	202° 58'	13"		1362.00	
2	271° 07'	59"		210.10	
2	259° 42'	45"		164.00	
2	295° 34'	53"		319.80	
2	237° 45'	39"		499.95	
2	283° 57'	32"		482.10	
The initial misclosure is:	DE	=	2.910		
	DN	=	-1.290		
Misclosure Le	Misclosure Length				
Misclosure Az	Misclosure Azimuth				

These are entered into the calculator as prompted. Once they have all been entered, an azimuth of -1 is entered, and the calculator shows the above misclosure data. Once this part is done, the calculator works through the adjustment of the sides, presenting the adjusted sides. The adjusted side values are:

Azimuth	Distance Correction	Distance
349° 47' 19"	0.4683	429.668
109° 22' 12"	-0.9666	475.033
202° 58' 13"	-0.4317	1361.568
271° 07' 59"	0.1719	210.272
259° 42' 45"	0.0930	164.093
295° 34' 53"	0.4393	320.239
237° 45' 39"	0.5639	500.514
283° 57' 32"	0.9761	483.076

Pressing R/S again, the calculator displays 1.314 as the sum of the corrections.

The final adjusted misclosure is:	DE = $1.600E - 11$
	DN = $2.500E - 11$
Misclosure Length	= 2.968E -11
Misclosure Azimuth	= 132° 37' 09"

The misclosure is now very much smaller than the finest level of distance measurement.

HP-35s Calculator Program Crandall's Adjustment for a Closed Traverse

Comments

An important point to note about Crandall's Adjustment is that because all the corrections are placed in the distances, any non-random errors (e.g., gross errors) in the angles will produce very large corrections in the distances. This is an indication that there may be gross errors in the angles, and this should be checked. Similarly, gross errors in the distances will also produce some large corrections.

In the second example, the original lengths were in many cases almost approximate in their apparent precision, while the misclosure was quite large, being about 1: 1,900. This led to the large corrections in the distances seen in the results.

Note also that in the second example, with the fixed side, the sum of the corrections was 1.314, well away from the zero. This is characteristic of adjusting traverses with fixed sides.

Storage Registers Used

- A Correlative A
- **B** Correlative B
- **C** Correction to side length
- **D** Current side length, intermediate result in adjustment calculation, adjusted length, misclosure length
- **E** Misclosure in departure, or the east-west direction.
- **I** Used to address the indirect registers, via (I)
- **J** Used to address the indirect registers, via (J)
- **K** Counter for the number of sides
- L Counter for the number of sides
- **N** Misclosure in latitude or the north-south direction
- **Q** Current side azimuth, azimuth of misclosure
- V Sum of corrections to the lengths

Statistical Registers: $\Sigma x = Sum of the latitudes$

 $\Sigma y = Sum of the departures$ $\Sigma x^2 = Sum of the squares of the latitudes of the sides to be adjusted$ $<math>\Sigma y^2 = Sum of the squares of the departures of the sides to be adjusted$ $<math>\Sigma xy = Sum of the products of the latitudes and departures of the sides to$ be adjusted

Labels Used

Label K Le	ength = 1058	Checksum $= 3I$)86
------------	--------------	-----------------	-----

Use the length (LN=) and Checksum (CK=) values to check if program was entered correctly. Use the sample computations to check proper operation after entry.

HP-35s Calculator Program Crandall's Adjustment for a Closed Traverse Flags Used

Flags 1 and 10 are used by this program. Flag 10 is set for this program, so that equations can be shown as prompts. Flag 1 is used to record the setting of Flag 10 before the program begins. At the end of the program, Flag 10 is reset to its original value, based on the value in Flag 1.

Reference

CRANDALL, Charles L., 1914. Geodesy and Least Squares. New York : John Wiley and Sons.